
Introduction

I Opération Binaire : 2 arguments de même type
Opérations arithmétiques, relations d’ordre (=,+, > etc.)

I Programmation OO : 2 objets de même classe
Tirer parti du polymorphisme

I D’où le terme de « méthode binaire »
I [Bruce et al., 1995] :

I Concept problématique dans l’approche objet traditionnelle
I Relation classe / type dans un contexte d’héritage

Contexte

La Hiérarchie Point

Point
x : int
y : int

+equal (p : Point) : Bool

ColorPoint
color : String

+equal (cp : ColorPoint) : Bool

C++ : Tentative #1
class Point
{

int x, y;

bool equal (Point& p)
{ return x == p.x && y == p.y; }

};

class ColorPoint : public Point
{

std::string color;

bool equal (ColorPoint& cp)
{ return color == cp.color && Point::equal (cp); }

};

C++ : Tentative #1

Échec
int main (int argc, char *argv[])
{

Point& p1 = * new ColorPoint (1, 2, "red");
Point& p2 = * new ColorPoint (1, 2, "green");

std::cout << p1.equal (p2) << std::endl;
// => True. #### Wrong !

}

I ColorPoint::equal masque Point::equal (statiquement)
I Seule Point::equal est vue depuis la classe de base
I On a besoin de la version exacte

C++ : Tentative #2
class Point
{

int x, y;

virtual bool equal (Point& p)
{ return x == p.x && y == p.y; }

};

class ColorPoint : public Point
{

std::string color;

virtual bool equal (ColorPoint& cp)
{ return color == cp.color && Point::equal (cp); }

};

C++ : Tentative #2

Échec
bool foo (Point& p1, Point& p2)
{

return p1.equal (p2);
}

Un ColorPoint Un Point

ColorPoint::equal
attends un ColorPoint…

…mais n’obtient
qu’un Point !

Contexte

La Hiérarchie Point

Point
x : int
y : int

+equal (p : Point) : Bool

ColorPoint
color : String

+equal (cp : ColorPoint) : Bool

Rappels : Covariance / Contravariance
I Règle Théorique :

I contravariance sur les arguments
I covariance sur les valeurs de retour

I En Pratique :
I invariance sur les arguments
I Ambiguïtés dues à la surcharge

I Remarque :
I Eiffel autorise la covariance sur les arguments
I Erreurs potentielles à l’exécution

I Analyse : [Castagna, 1995]
I Manque d’expressivité

sous-typage (par sous-classage) 6= spécialisation
I Défaut du modèle object classique

Absence des multi-méthodes

CLOS : Tentative #1

(defgeneric point= (a b))

(defclass point ()
((x :initarg :x :reader point-x)
(y :initarg :y :reader point-y)))

(defmethod point= ((a point) (b point))
(and (= (point-x a) (point-x b))

(= (point-y a) (point-y b))))

(defclass color-point (point)
((color :initarg :color :reader point-color)))

(defmethod point= ((a color-point) (b color-point))
(and (string= (point-color a) (point-color b))

(call-next-method)))

CLOS : Tentative #1

Succès
(let ((p1 (make-point :x 1 :y 2))

(p2 (make-point :x 1 :y 2))
(cp1 (make-color-point :x 1 :y 2 :color "red"))
(cp2 (make-color-point :x 1 :y 2 :color "green")))

(values (point= p1 p2)
(point= cp1 cp2)))

;; => (T NIL)

I Sélection de la méthode appropriée sur les 2 arguments
Multiple dispatch

I Syntaxe d’appel de fonction plus satisfaisant
p1.equal(p2) or p2.equal(p1) ?

I « Fonction binaire »

Rappel : Combinaisons de Méthodes
I Éviter la duplication de code :

I C++ : Point::equal()
I CLOS : (call-next-method)

I Méthodes Applicables :
I Toutes les méthodes compatibles avec les classes des arguments
I Classées par spécificité décroissante
I call-next-method exécute la méthode suivante dans cet ordre

I Combinaisons de méthodes :
I Appeler plusieurs méthodes applicables

Pas seulement la plus spécifique
I Combinaisions prédéfinies : and, or, progn etc.
I Programmable

La Combinaison and

(defgeneric point= (a b)
(:method-combination and))

(defmethod point= and ((a point) (b point))
(and (= (point-x a) (point-x b))

(= (point-y a) (point-y b))))

(defmethod point= and ((a color-point) (b color-point))
(and (call-next-method)

(string= (point-color a) (point-color b))))

I Note : le dispatch de CLOS est programmable

Utilisation Non Contractuelle

Succès non mérité
(let ((p (make-point :x 1 :y 2))

(cp (make-color-point :x 1 :y 2 :color "red")))
(point= p cp))

;; => T #### Wrong !

I (point= <point> <point>) est applicable
Un color-point est un point

I Se prémunir contre une telle situation

Introspection dans CLOS

(assert (eq (class-of a) (class-of b)))

I Quand vérifier ?
I Chaque méthode : duplication
I Dernière méthode : pas efficace / pas toujours possible
I :before : pas disponible hors combinaison standard
I Nouvelle combination : pas le lieu
I :around : oui mais…

I Où vérifier ? (meilleure question)
I Nulle part dans le code de point=
I Lié au concept de fonction binaire, pas à point=

I Exprimer le concept de fonction binaire lui-même
I Les fonctions binaires sont des fonctions génériques particulières

Une Méta-Classe de Fonctions Binaires

I Fonctions génériques : standard-generic-function
I Objets-fonctions : funcallable-standard-class

(defclass binary-function (standard-generic-function)
()
(:metaclass funcallable-standard-class))

(defmacro defbinary (function-name lambda-list &rest options)
`(defgeneric ,function-name ,lambda-list
(:generic-function-class binary-function)
,@options))

I Remarque : le concept impose une lambda-list particulière…

Appels Génériques

I compute-discriminating-function :
calcul, classement et exécution des méthodes
applicables

I compute-applicable-methods-using-classes :
calcul et classement des méthodes applicables

I compute-effective-method :
Application des méthodes (combinées) aux
arguments

I Note : fonctions génériques spécialisables

Protocole
generic call

comp-dscr-func

comp-app-meth-usg-cls

comp-eff-meth

Couche Utilisateur
Couche 3

Application aux Fonctions Binaires

(defmethod compute-applicable-methods-using-classes
:before ((binary-function binary-function) classes)

(assert (= (length classes) 2))
(assert (apply #'eq classes)))

Implémentation Non-Contractuelle

;; #### WRONG SPEC!!
(defmethod point= ((p1 point-3d) (p2 point-2d))

#| ... |#)

I Précédemment : protection contre les appels illégaux
I Question : protection contre les implémentations illégales ?

Définition de Méthodes

I ensure-method :
création et ajout de la nouvelle méthode

I make-method :
création du méta-objet

I add-method :
mise à jour de la fonction générique

I Note : fonctions génériques spécialisables

Protocole
defmethod

ensure-method

make-method

add-method

Couche 1
Couche 2
Couche 3

Application aux Fonctions Binaires

(defmethod add-method :before ((bf binary-function) method)
(let ((method-specializers (method-specializers method)))
(assert (= (length method-specializers) 2))
(assert (apply #'eq method-specializers))))

I method : méta-objet (cf. standard-method)
I method-specializers : accesseur

Implémentation Manquante

I Chaque sous-classe de point doit spécialiser point=
I Vérification tardive : lors d’un appel générique

préserver la haute dynamicité
I « Binaire-complétude » : il existe une méthode primaire applicable pour chaque classe de

la hiérarchie

(defmethod compute-applicable-methods-using-classes
:around ((bf binary-function) classes)

(multiple-value-bind (methods ok) (call-next-method)
;; 1. Check for binary completeness on METHODS
;; 2. If everything goes well, simply return
;; what we got from CALL-NEXT-METHOD
(values methods ok)))

Mise en oeuvre
(loop :for class :in (class-precedence-list (car classes))

:until (eq class (find-class 'standard-object))
:do (assert

(find-if
(lambda (method)

(let ((qualifiers (method-qualifiers method)))
(and (equal (method-specializers method)

(list class class))
(not (member :around qualifiers))
(not (member :before qualifiers))
(not (member :after qualifiers)))))

methods)))

I class-precedence-list : accesseur
I method-qualifiers : accesseur

Conclusion
I Méthodes binaires : un concept très simple

I Difficile à implémenter dans un système OO classique
I Trivial avec des multi-méthodes

I Grâce à CLOS :
I Validation de l’utilisation
I Validation de l’implémentation

I Grâce au MOP :
I Concept explicite
I Extension du système OO de base

